/*******************************************************************************
 *  Copyright 2012-2022 Esri
 *
 *  Licensed under the Apache License, Version 2.0 (the "License");
 *  you may not use this file except in compliance with the License.
 *  You may obtain a copy of the License at
 *
 *  http://www.apache.org/licenses/LICENSE-2.0
 *
 *  Unless required by applicable law or agreed to in writing, software
 *  distributed under the License is distributed on an "AS IS" BASIS,
 *  WITHOUT WARRANTIES OR CONDITIONS OF ANY KIND, either express or implied.
 *  See the License for the specific language governing permissions and
 *  limitations under the License.
 ******************************************************************************/

import QtQuick
import QtQuick.Controls
import Esri.ArcGISRuntime
import Esri.ArcGISRuntime.Toolkit
import "tools.js" as T

Item {
    Drawer {
        id: drawer
        height: parent.height
        implicitWidth: 300
        clip: true
        dragMargin: 0
        ListView {
            id: viewList
            anchors.fill: parent
            model: ListModel {
                id: listModel
                Component.onCompleted: {
                    for (let tool of T.TOOLS) {
                        listModel.append(tool);
                    }
                    viewList.currentIndex = 0;
                }
            }
            delegate: ItemDelegate {
                text: name
                highlighted: ListView.isCurrentItem
                width: parent.width
                onClicked: {
                    viewList.currentIndex = index;
                    drawer.close();
                }
            }
        }
    }

    Connections {
        target: content.item
        function onShowToolsButtonPressed() {
            drawer.open()
        }
    }

    Loader {
        id: content
        anchors.fill: parent
        focus: true
        source: viewList.currentIndex >= 0 ? listModel.get(viewList.currentIndex).url : ""
        onLoaded: {item.handlesOwnAuthentication = listModel.get(viewList.currentIndex).handlesOwnAuthentication ?? false;}
    }
}
